
----------------------------------------------------------------------------

*&---------------------------------------------------------------------*

*& T-delay Process Chain block                                    *

*&---------------------------------------------------------------------*

*& Is executed till achieving P_STIME time                   *

*&---------------------------------------------------------------------*

REPORT  ZBI_T_DELAY .

parameters : P_STIME type T ,                " Start Time

                      P_MDNGHT type C as checkbox .   " Over midnight

data : iseconds type I .

data : to_wait(1) type C ,

       to_exit(1) type C .

data : block_start type T .

data : ajusted_time(6) type N .

start-of-selection .

  block_start = sy-uzeit .      " T_DELAY block start time

  iseconds =  60 .               " 1 Minute Debug Loop interval

  if P_STIME is initial .

    P_STIME = sy-uzeit .

  endif .

  clear to_exit .

  while to_exit is initial .

    perform check_midnight .

    perform check_start_time changing to_wait.

    if not to_wait is initial .

      CALL FUNCTION 'RSSM_SLEEP_DEBUG'   " Debug Loop

        EXPORTING

          I_SECONDS                = iseconds

          I_ALWAYS                 = 'X' .

    else.

      to_exit = 'X' .

    endif .

  endwhile .

*&---------------------------------------------------------------------*

*&      Form  check_start_time

*&---------------------------------------------------------------------*

FORM check_start_time  CHANGING to_wait.

data : curr_time(6) type N .

  clear to_wait .

  curr_time = sy-uzeit .

  if curr_time < ajusted_time .

    to_wait = 'X'.

  endif .

ENDFORM.                    " check_start_time

*&---------------------------------------------------------------------*

*&      Form  check_midnight

*&---------------------------------------------------------------------*

FORM check_midnight .

  ajusted_time = P_STIME .

  if P_MDNGHT = 'X' .

    if block_start > P_STIME .

      if  sy-uzeit <= '23:59:59'.

        add 240000 to ajusted_time .

      endif .

    endif .

  endif .

ENDFORM.                    " check_midnight
-----------------------------------------------------------------------------

T-delay block — ABAP code implementation
------------------------------------------------------------------------------
*&---------------------------------------------------------------------*

*& PC-Delay Process Chain Block                               *

*&---------------------------------------------------------------------*

*&  Executed while S_PC Process Chains                    *

*&  Started P_BCKDAY days before are active             *

*&  Canceled after P_W_MAX minutes (0 - not canceled)   

*&---------------------------------------------------------------------*

REPORT  ZBI_PC_DELAY .

tables: RSPCLOGCHAIN , RSPCPROCESSLOG .

select-options S_PC for RSPCLOGCHAIN-CHAIN_ID .

parameters : P_BCKDAY type I ,

             P_W_MAX  type I ,

             P_0K RADIOBUTTON GROUP RADI default 'X',

             P_CANCEL RADIOBUTTON GROUP RADI ,

             P_ANY RADIOBUTTON GROUP RADI .

data : fdate like sy-datum .

data : CHAIN_ID  like RSPCLOGCHAIN-CHAIN_ID,

       ACTUAL_STATE like RSPCPROCESSLOG-ACTUAL_STATE ,

       DATUM like RSPCLOGCHAIN-DATUM.

data : iseconds type I .

data : is_active(1) type C ,

       to_exit(1) type C ,

       is_max(1) type C .

data : f_tst TYPE timestampl ,

       t_tst TYPE timestampl ,

       tmin  TYPE I.

ranges : R_STATE for RSPCPROCESSLOG-ACTUAL_STATE .

start-of-selection .

  fdate = sy-datum - P_BCKDAY .

  iseconds =  60 .   " 1 Minute

  if P_0K = 'X' or P_ANY = 'X' .

    R_STATE-LOW = 'G' .

     R_STATE-SIGN = 'I' .

      R_STATE-OPTION = 'EQ' .

       append R_STATE .

    R_STATE-LOW = 'F' .

     R_STATE-SIGN = 'I' .

      R_STATE-OPTION = 'EQ' .

       append R_STATE .

  endif .

  if P_CANCEL = 'X' or P_ANY = 'X' .

    R_STATE-LOW = 'X' .

     R_STATE-SIGN = 'I' .

      R_STATE-OPTION = 'EQ' .

       append R_STATE .

    R_STATE-LOW = 'R' .

     R_STATE-SIGN = 'I' .

      R_STATE-OPTION = 'EQ' .

       append R_STATE .

  endif .

  GET TIME STAMP FIELD f_tst.

  clear: to_exit , is_max .

  while to_exit is initial .

    perform check_predecessor_pc changing is_active .

    if not is_active is initial .

      CALL FUNCTION 'RSSM_SLEEP_DEBUG'      " Debug Loop

        EXPORTING

          I_SECONDS                = iseconds

          I_ALWAYS                 = 'X' .

      if not P_W_MAX is initial . " check achiving Max execution time

        perform get_time_diff changing tmin .

        IF tmin > P_W_MAX.

          to_exit = 'X' .

          is_max = 'X' .         " Max execution time is achived

        endif .

      endif .

    else.

      to_exit = 'X' .

    endif .

  endwhile .

  if  not is_max is initial.

    MESSAGE E090(RSPC) WITH CHAIN_ID ACTUAL_STATE.

*   Previous run &1 not yet completed (status &2)

  endif .

*&---------------------------------------------------------------------*

*& Check if predecessor PCs are still active

*&---------------------------------------------------------------------*

FORM check_predecessor_pc  CHANGING is_active.

data : LOG_ID like RSPCLOGCHAIN-LOG_ID,

       CHAIN_ID like RSPCLOGCHAIN-CHAIN_ID .

data : begin of t_last_PC occurs 0 ,

         CHAIN_ID like RSPCLOGCHAIN-CHAIN_ID ,

         LOG_ID   like RSPCLOGCHAIN-LOG_ID ,

       end of t_last_PC .

       data : E_STATUS TYPE  RSPC_STATE .

ranges : R_LOG_ID for RSPCLOGCHAIN-LOG_ID .

  clear is_active .

*--- Get the last PC Log

  select *

     from RSPCLOGCHAIN

     where DATUM  >= fdate

       and CHAIN_ID in S_PC

     order by CHAIN_ID DATUM ZEIT .

    read table  t_last_PC with key CHAIN_ID = RSPCLOGCHAIN-CHAIN_ID .

    move-corresponding RSPCLOGCHAIN to t_last_PC .

    if sy-subrc is initial .

      modify t_last_PC index sy-tabix.

    else .

      append t_last_PC .

    endif .

  endselect .

*--- Check with S_PC list !?

  read table t_last_PC index 1 .

  if sy-tfill = 0 .    " No Data

    is_active = 'X'.

  else .

    loop at t_last_PC .

      CALL FUNCTION 'RSPC_API_CHAIN_GET_STATUS'

        EXPORTING

          I_CHAIN     = t_last_PC-CHAIN_ID

          I_LOGID     = t_last_PC-LOG_ID

        IMPORTING

          E_STATUS    = E_STATUS .

      if not E_STATUS in R_STATE .

        is_active = 'X'.

        exit .

      endif .

    endloop .

  endif.

ENDFORM.

*&---------------------------------------------------------------------*

*&      Form  get_time_diff

*&---------------------------------------------------------------------*

FORM get_time_diff  CHANGING t_min.

data : l_sec          TYPE i,

       l_milisec      TYPE i,

       l_day          TYPE i,

       t_sec(16)      TYPE P DECIMALS 6 .

  GET TIME STAMP FIELD t_tst.

  CALL FUNCTION 'RSSM_SUBSTRACT_TIMESTAMPS'

    EXPORTING

      TIMESTAMP1       = f_tst

      TIMESTAMP2       = t_tst

    IMPORTING

      SECS             = l_sec

      DAYS             = l_day

      MILISECS         = l_milisec .

t_sec = ( l_day * 86400 ) + l_sec + ( l_milisec / 10000000 ).

t_min = t_sec / 60.

ENDFORM.                    " get_time_diff

------------------------------------------------------------------------------

